NAG Library Routine Document

F08HAF (DSBEV)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08HAF (DSBEV) computes all the eigenvalues and, optionally, all the eigenvectors of a real n by n symmetric band matrix A of bandwidth $(2k_d + 1)$.

2 Specification

```
SUBROUTINE F08HAF (JOBZ, UPLO, N, KD, AB, LDAB, W, Z, LDZ, WORK, INFO)

INTEGER             N, KD, LDAB, LDZ, INFO
REAL (KIND=nag_wp) AB(LDAB,*), W(N), Z(LDZ,*), WORK(3*N-2)
CHARACTER(1)        JOBZ, UPLO
```

The routine may be called by its LAPACK name ***dsbev***.

3 Description

The symmetric band matrix A is first reduced to tridiagonal form, using orthogonal similarity transformations, and then the QR algorithm is applied to the tridiagonal matrix to compute the eigenvalues and (optionally) the eigenvectors.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: JOBZ – CHARACTER(1) *Input*

On entry: indicates whether eigenvectors are computed.

JOBZ = 'N'
Only eigenvalues are computed.

JOBZ = 'V'
Eigenvalues and eigenvectors are computed.

Constraint: JOBZ = 'N' or 'V'.

2: UPLO – CHARACTER(1) *Input*

On entry: if UPLO = 'U', the upper triangular part of A is stored.

If UPLO = 'L', the lower triangular part of A is stored.

Constraint: UPLO = 'U' or 'L'.

3: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

4: KD – INTEGER *Input*

On entry: if UPLO = 'U', the number of superdiagonals, k_d, of the matrix A.

If UPLO = 'L', the number of subdiagonals, k_d, of the matrix A.

Constraint: KD ≥ 0.

5: AB(LDAB,$*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array AB must be at least $\max(1, \text{N})$.

On entry: the upper or lower triangle of the n by n symmetric band matrix A.

The matrix is stored in rows 1 to $k_d + 1$, more precisely,

if UPLO = 'U', the elements of the upper triangle of A within the band must be stored with element A_{ij} in AB$(k_d + 1 + i - j, j)$ for $\max(1, j - k_d) \leq i \leq j$;

if UPLO = 'L', the elements of the lower triangle of A within the band must be stored with element A_{ij} in AB$(1 + i - j, j)$ for $j \leq i \leq \min(n, j + k_d)$.

On exit: AB is overwritten by values generated during the reduction to tridiagonal form.

The first superdiagonal or subdiagonal and the diagonal of the tridiagonal matrix T are returned in AB using the same storage format as described above.

6: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F08HAF (DSBEV) is called.

Constraint: LDAB $\geq$ KD + 1.

7: W(N) – REAL (KIND=nag_wp) array *Output*

On exit: the eigenvalues in ascending order.

8: Z(LDZ,$*$) – REAL (KIND=nag_wp) array *Output*

Note: the second dimension of the array Z must be at least $\max(1, \text{N})$ if JOBZ = 'V', and at least 1 otherwise.

On exit: if JOBZ = 'V', Z contains the orthonormal eigenvectors of the matrix A, with the ith column of Z holding the eigenvector associated with W(i).

If JOBZ = 'N', Z is not referenced.

9: LDZ – INTEGER *Input*

On entry: the first dimension of the array Z as declared in the (sub)program from which F08HAF (DSBEV) is called.

Constraints:

if JOBZ = 'V', LDZ $\geq \max(1, \text{N})$;
otherwise LDZ ≥ 1.

10: WORK($3 \times$ N $- 2$) – REAL (KIND=nag_wp) array *Workspace*

11: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, the algorithm failed to converge; i off-diagonal elements of an intermediate tridiagonal form did not converge to zero.

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix $(A + E)$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the ***machine precision***. See Section 4.7 of Anderson *et al.* (1999) for further details.

8 Further Comments

The total number of floating point operations is proportional to n^3 if JOBZ = 'V' and is proportional to $k_d n^2$ otherwise.

The complex analogue of this routine is F08HNF (ZHBEV).

9 Example

This example finds all the eigenvalues and eigenvectors of the symmetric band matrix

$$A = \begin{pmatrix} 1 & 2 & 3 & 0 & 0 \\ 2 & 2 & 3 & 4 & 0 \\ 3 & 3 & 3 & 4 & 5 \\ 0 & 4 & 4 & 4 & 5 \\ 0 & 0 & 5 & 5 & 5 \end{pmatrix},$$

together with approximate error bounds for the computed eigenvalues and eigenvectors.

9.1 Program Text

```
    Program f08hafe

!     F08HAF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: ddisna, dsbev, nag_wp, x02ajf, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter                :: nin = 5, nout = 6
      Character (1), Parameter          :: uplo = 'U'
!     .. Local Scalars ..
      Real (Kind=nag_wp)                :: eerrbd, eps
      Integer                           :: i, ifail, info, j, kd, ldab, ldz, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable   :: ab(:,:), rcondz(:), w(:), work(:),   &
                                           z(:,:), zerrbd(:)
!     .. Intrinsic Procedures ..
      Intrinsic                         :: abs, max, min
!     .. Executable Statements ..
```

```
      Write (nout,*) 'F08HAF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, kd
      ldab = kd + 1
      ldz = n
      Allocate (ab(ldab,n),rcondz(n),w(n),work(3*n-2),z(ldz,n),zerrbd(n))

!     Read the upper or lower triangular part of the symmetric band
!     matrix A from data file

      If (uplo=='U') Then
        Read (nin,*)((ab(kd+1+i-j,j),j=i,min(n,i+kd)),i=1,n)
      Else If (uplo=='L') Then
        Read (nin,*)((ab(1+i-j,j),j=max(1,i-kd),i),i=1,n)
      End If

!     Solve the band symmetric eigenvalue problem

!     The NAG name equivalent of dsbev is f08haf
      Call dsbev('Vectors',uplo,n,kd,ab,ldab,w,z,ldz,work,info)

      If (info==0) Then

!       Print solution

        Write (nout,*) 'Eigenvalues'
        Write (nout,99999) w(1:n)
        Flush (nout)

!       Standardize the eigenvectors so that first elements are non-negative.
        Do i = 1, n
          If (z(1,i)<0.0_nag_wp) z(1:n,i) = -z(1:n,i)
        End Do

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04caf('General',' ',n,n,z,ldz,'Eigenvectors',ifail)

!       Get the machine precision, EPS and compute the approximate
!       error bound for the computed eigenvalues.  Note that for
!       the 2-norm, max( abs(W(i)) ) = norm(A), and since the
!       eigenvalues are returned in ascending order
!       max( abs(W(i)) ) = max( abs(W(1)), abs(W(n)))

        eps = x02ajf()
        eerrbd = eps*max(abs(w(1)),abs(w(n)))

!       Call DDISNA (F08FLF) to estimate reciprocal condition
!       numbers for the eigenvectors
        Call ddisna('Eigenvectors',n,n,w,rcondz,info)

!       Compute the error estimates for the eigenvectors

        Do i = 1, n
          zerrbd(i) = eerrbd/rcondz(i)
        End Do

!       Print the approximate error bounds for the eigenvalues
!       and vectors

        Write (nout,*)
        Write (nout,*) 'Error estimate for the eigenvalues'
        Write (nout,99998) eerrbd
        Write (nout,*)
        Write (nout,*) 'Error estimates for the eigenvectors'
        Write (nout,99998) zerrbd(1:n)
      Else
        Write (nout,99997) 'Failure in DSBEV. INFO =', info
```

```
      End If

99999 Format (3X,(8F8.4))
99998 Format (4X,1P,6E11.1)
99997 Format (1X,A,I4)
    End Program f08hafe
```

9.2 Program Data

```
F08HAF Example Program Data

  5    2                   :Values of N and KD

  1.0  2.0  3.0
       2.0  3.0  4.0
            3.0  4.0  5.0
                 4.0  5.0
                      5.0  :End of matrix A
```

9.3 Program Results

```
 F08HAF Example Program Results

 Eigenvalues
    -3.2474 -2.6633  1.7511  4.1599 14.9997
 Eigenvectors
          1       2       3       4       5
 1   0.0394  0.6238  0.5635  0.5165  0.1582
 2   0.5721 -0.2575 -0.3896  0.5955  0.3161
 3  -0.4372 -0.5900  0.4008  0.1470  0.5277
 4  -0.4424  0.4308 -0.5581 -0.0470  0.5523
 5   0.5332  0.1039  0.2421 -0.5956  0.5400

 Error estimate for the eigenvalues
        1.7E-15

 Error estimates for the eigenvectors
        2.9E-15    2.9E-15    6.9E-16    6.9E-16    1.5E-16
```